

# NAG Library Routine Document

## G04BBF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

### 1 Purpose

G04BBF computes the analysis of variance and treatment means and standard errors for a randomized block or completely randomized design.

### 2 Specification

```

SUBROUTINE G04BBF (N, Y, IBLOCK, NT, IT, GMEAN, BMEAN, TMEAN, TABL,      &
                  LDATABL, C, LDC, IREP, R, EF, TOL, IRDF, WK, IFAIL)
INTEGER            N, IBLOCK, NT, IT(*), LDATABL, LDC, IREP(NT), IRDF,  &
                  IFAIL
REAL (KIND=nag_wp) Y(N), GMEAN, BMEAN(abs(IBLOCK)), TMEAN(NT),      &
                  TABL(LDATABL,5), C(LDC,NT), R(N), EF(NT), TOL,      &
                  WK(3*NT)

```

### 3 Description

In a completely randomized design, experimental material is divided into a number of units, or plots, to which a treatment can be applied. In a randomized block design the units are grouped into blocks so that the variation within blocks is less than the variation between blocks. If every treatment is applied to one plot in each block it is a complete block design. If there are fewer plots per block than treatments then the design will be an incomplete block design and may be balanced or partially balanced.

For a completely randomized design, with  $t$  treatments and  $n_t$  plots per treatment, the linear model is

$$y_{ij} = \mu + \tau_j + e_{ij}, \quad j = 1, 2, \dots, t \text{ and } i = 1, 2, \dots, n_j,$$

where  $y_{ij}$  is the  $i$ th observation for the  $j$ th treatment,  $\mu$  is the overall mean,  $\tau_j$  is the effect of the  $j$ th treatment and  $e_{ij}$  is the random error term. For a randomized block design, with  $t$  treatments and  $b$  blocks of  $k$  plots, the linear model is

$$y_{ij(l)} = \mu + \beta_i + \tau_l + e_{ij}, \quad i = 1, 2, \dots, b, j = 1, 2, \dots, k \text{ and } l = 1, 2, \dots, t,$$

where  $\beta_i$  is the effect of the  $i$ th block and the  $ij(l)$  notation indicates that the  $l$ th treatment is applied to the  $j$ th plot in the  $i$ th block.

The completely randomized design gives rise to a one-way analysis of variance. The treatments do not have to be equally replicated, i.e., do not have to occur the same number of times. First the overall mean,  $\hat{\mu}$ , is computed and subtracted from the observations to give  $y'_{ij} = y_{ij} - \hat{\mu}$ . The estimated treatment effects,  $\hat{\tau}_j$  are then computed as the treatment means of the mean adjusted observations,  $y'_{ij}$ , and the treatment sum of squares can be computed from the sum of squares of the treatment totals of the  $y'_{ij}$  divided by the number of observations per treatment total,  $n_j$ . The final residuals are computed as  $r_{ij} = y'_{ij} - \hat{\tau}_j$ , and, from the residuals, the residual sum of squares is calculated.

For the randomized block design the mean is computed and subtracted from the observations to give  $y'_{ij(l)} = y_{ij(l)} - \hat{\mu}$ . The estimated block effects, ignoring treatment effects,  $\hat{\beta}_i$ , are then computed using the block means of the  $y'_{ij(l)}$  and the unadjusted sum of squares computed as the sum of squared block totals for the  $y'_{ij(l)}$  divided by number of plots per block,  $k$ . The block adjusted observations are then computed as  $y''_{ij(l)} = y'_{ij(l)} - \hat{\beta}_i$ . In the case of the complete block design, with the same replication for each treatment within each block, the blocks and treatments are orthogonal, and so the treatment effects are estimated as the treatment means of the block adjusted observations,  $y''_{ij(l)}$ . The treatment sum of

squares is computed as the sum of squared treatment totals of the  $y''_{ij(l)}$  divided by the number of replicates to the treatments,  $r = bk/t$ . Finally the residuals, and hence the residual sum of squares, are given by  $r_{ij(l)} = y''_{ij(l)} - \hat{\tau}_l$ .

For a design without the same replication for each treatment within each block the treatments and the blocks will not be orthogonal, so the treatments adjusted for blocks need to be computed. The adjusted treatment effects are found as the solution to the equations

$$(R - NN^T/k)\hat{\tau} = q,$$

where  $q$  is the vector of the treatment totals for block adjusted observations,  $y''_{ij(l)}$ ,  $R$  is a diagonal matrix with  $R_{ll}$  equal to the number of times the  $l$ th treatment is replicated, and  $N$  is the  $t$  by  $b$  incidence matrix, with  $N_{lj}$  equal to the number of times treatment  $l$  occurs in block  $j$ . The solution to the equations can be written as

$$\hat{\tau} = \Omega q$$

where  $\Omega$  is a generalized inverse of  $(R - NN^T/k)$ . The solution is found from the eigenvalue decomposition of  $(R - NN^T/k)$ . The residuals are first calculated by subtracting the estimated treatment effects from the block adjusted observations to give  $r'_{ij(l)} = y''_{ij(l)} - \hat{\tau}_l$ . However, since only the unadjusted block effects have been removed and blocks and treatments are not orthogonal, the block means of the  $r'_{ij(l)}$  have to be subtracted to give the correct residuals,  $r_{ij(l)}$  and residual sum of squares.

The mean squares are computed as the sum of squares divided by the degrees of freedom. The degrees of freedom for the unadjusted blocks is  $b - 1$ , for the completely randomized and the complete block designs the degrees of freedom for the treatments is  $t - 1$ . In the general case the degrees of freedom for treatments is the rank of the matrix  $\Omega$ . The  $F$ -statistic given by the ratio of the treatment mean square to the residual mean square tests the hypothesis

$$H_0 : \tau_1 = \tau_2 = \dots = \tau_t = 0.$$

The standard errors for the difference in treatment effects, or treatment means, for the completely randomized or the complete block designs, are given by:

$$\text{se}(\tau_j - \tau_{j*}) = \left( \frac{1}{n_j} + \frac{1}{n_{j*}} \right) s^2$$

where  $s^2$  is the residual mean square and  $n_j = n_{j*} = b$  in the complete block design. In the general case the variances of the treatment effects are given by

$$\text{var}(\tau) = \Omega s^2$$

from which the appropriate standard errors of the difference between treatment effects or the difference between adjusted means can be calculated.

In the complete block design all the information on the treatment effects is given by the within block analysis. In other designs there may be a loss of information due to the non-orthogonality of treatments and blocks. The efficiency of the within block analysis in these cases is given by the (canonical) efficiency factors, these are the nonzero eigenvalues of the matrix  $(R - NN^T/k)$ , divided by the number of replicates in the case of equal replication, or by the mean of the number of replicates in the unequally replicated case, see John (1987). If more than one eigenvalue is zero then the design is said to be disconnected and some treatments can only be compared using a between block analysis.

## 4 References

Cochran W G and Cox G M (1957) *Experimental Designs* Wiley

Davis O L (1978) *The Design and Analysis of Industrial Experiments* Longman

John J A (1987) *Cyclic Designs* Chapman and Hall

John J A and Quenouille M H (1977) *Experiments: Design and Analysis* Griffin

Searle S R (1971) *Linear Models* Wiley

## 5 Arguments

- 1: N – INTEGER *Input*  
*On entry:* the number of observations.  
*Constraint:*  $N \geq 2$  and if  $\text{abs}(\text{IBLOCK}) \geq 2$ , N must be a multiple of  $\text{abs}(\text{IBLOCK})$ .
- 2: Y(N) – REAL (KIND=nag\_wp) array *Input*  
*On entry:* the observations in the order as described by IBLOCK and NT.
- 3: IBLOCK – INTEGER *Input*  
*On entry:* indicates the block structure.  
 $\text{abs}(\text{IBLOCK}) \leq 1$   
 There are no blocks, i.e., it is a completely randomized design.  
 $\text{IBLOCK} \geq 2$   
 There are IBLOCK blocks and the data should be input by blocks, i.e., Y must contain the observations for block 1 followed by the observations for block 2, etc.  
 $\text{IBLOCK} \leq -2$   
 There are  $\text{abs}(\text{IBLOCK})$  blocks and the data is input in parallel with respect to blocks, i.e., Y(1) must contain the first observation for block 1, Y(2) must contain the first observation for block 2...Y( $\text{abs}(\text{IBLOCK})$ ) must contain the first observation for block  $\text{abs}(\text{IBLOCK})$ , Y( $\text{abs}(\text{IBLOCK} + 1)$ ) must contain the second observation for block 1, etc.  
*Constraint:* IBLOCK = 1, 2 or -2.
- 4: NT – INTEGER *Input*  
*On entry:* the number of treatments. If only blocks are required in the analysis then set NT = 1.  
*Constraints:*  
 if  $\text{abs}(\text{IBLOCK}) \geq 2$ ,  $\text{NT} \geq 1$ ;  
 otherwise  $\text{NT} \geq 2$ .
- 5: IT(\*) – INTEGER array *Input*  
**Note:** the dimension of the array IT must be at least N if  $\text{NT} \geq 2$ , and at least 1 otherwise.  
*On entry:* IT(*i*) indicates which of the NT treatments plot *i* received, for  $i = 1, 2, \dots, N$ .  
 If NT = 1, IT is not referenced.  
*Constraint:*  $1 \leq \text{IT}(i) \leq \text{NT}$ , for  $i = 1, 2, \dots, N$ .
- 6: GMEAN – REAL (KIND=nag\_wp) *Output*  
*On exit:* the grand mean,  $\hat{\mu}$ .
- 7: BMEAN( $\text{abs}(\text{IBLOCK})$ ) – REAL (KIND=nag\_wp) array *Output*  
*On exit:* if  $\text{abs}(\text{IBLOCK}) \geq 2$ , BMEAN(*j*) contains the mean for the *j*th block,  $\hat{\beta}_j$ , for  $j = 1, 2, \dots, b$ .
- 8: TMEAN(NT) – REAL (KIND=nag\_wp) array *Output*  
*On exit:* if  $\text{NT} \geq 2$ , TMEAN(*l*) contains the (adjusted) mean for the *l*th treatment,  $\hat{\mu}^* + \hat{\tau}_l$ , for  $l = 1, 2, \dots, t$ , where  $\hat{\mu}^*$  is the mean of the treatment adjusted observations,  $y_{ij(l)} - \hat{\tau}_l$ .

- 9: TABL(LDTABL, 5) – REAL (KIND=nag\_wp) array *Output*  
*On exit:* the analysis of variance table. Column 1 contains the degrees of freedom, column 2 the sum of squares, and where appropriate, column 3 the mean squares, column 4 the  $F$ -statistic and column 5 the significance level of the  $F$ -statistic. Row 1 is for Blocks, row 2 for Treatments, row 3 for Residual and row 4 for Total. Mean squares are computed for all but the Total row;  $F$ -statistics and significance are computed for Treatments and Blocks, if present. Any unfilled cells are set to zero.
- 10: LDTABL – INTEGER *Input*  
*On entry:* the first dimension of the array TABL as declared in the (sub)program from which G04BBF is called.  
*Constraint:* LDTABL  $\geq 4$ .
- 11: C(LDC, NT) – REAL (KIND=nag\_wp) array *Output*  
*On exit:* if NT  $\geq 2$ , the upper triangular part of C contains the variance-covariance matrix of the treatment effects, the strictly lower triangular part contains the standard errors of the difference between two treatment effects (means), i.e.,  $C(i, j)$  contains the covariance of treatment  $i$  and  $j$  if  $j \geq i$  and the standard error of the difference between treatment  $i$  and  $j$  if  $j < i$ , for  $i = 1, 2, \dots, t$  and  $j = 1, 2, \dots, t$ .
- 12: LDC – INTEGER *Input*  
*On entry:* the first dimension of the array C as declared in the (sub)program from which G04BBF is called.  
*Constraint:* LDC  $\geq$  NT.
- 13: IREP(NT) – INTEGER array *Output*  
*On exit:* if NT  $\geq 2$ , the treatment replications,  $R_{ll}$ , for  $l = 1, 2, \dots, NT$ .
- 14: R(N) – REAL (KIND=nag\_wp) array *Output*  
*On exit:* the residuals,  $r_i$ , for  $i = 1, 2, \dots, N$ .
- 15: EF(NT) – REAL (KIND=nag\_wp) array *Output*  
*On exit:* if NT  $\geq 2$ , the canonical efficiency factors.
- 16: TOL – REAL (KIND=nag\_wp) *Input*  
*On entry:* the tolerance value used to check for zero eigenvalues of the matrix  $\Omega$ . If TOL = 0.0 a default value of  $10^{-5}$  is used.  
*Constraint:* TOL  $\geq 0.0$ .
- 17: IRDF – INTEGER *Input*  
*On entry:* an adjustment to the degrees of freedom for the residual and total.  
 IRDF  $\geq 1$   
 The degrees of freedom for the total is set to  $N - IRDF$  and the residual degrees of freedom adjusted accordingly.  
 IRDF = 0  
 The total degrees of freedom for the total is set to  $N - 1$ , as usual.  
*Constraint:* IRDF  $\geq 0$ .

18: WK(3 × NT) – REAL (KIND=nag\_wp) array *Workspace*

19: IFAIL – INTEGER *Input/Output*

*On entry:* IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, because for this routine the values of the output arguments may be useful even if IFAIL ≠ 0 on exit, the recommended value is -1. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

*On exit:* IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

## 6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

**Note:** G04BBF may return useful information for one or more of the following detected errors or warnings.

Errors or warnings detected by the routine:

IFAIL = 1

On entry,  $N < 2$ ,  
 or  $NT \leq 0$ ,  
 or  $NT = 1$  and  $\text{abs}(\text{IBLOCK}) \leq 1$ ,  
 or  $\text{LDTABL} < 4$ ,  
 or  $\text{LDC} < NT$ ,  
 or  $\text{TOL} < 0.0$ ,  
 or  $\text{IRDF} < 0$ .

IFAIL = 2

On entry,  $\text{abs}(\text{IBLOCK}) \geq 2$  and  $N$  is not a multiple of  $\text{abs}(\text{IBLOCK})$ .

IFAIL = 3

On entry,  $\text{IT}(i) < 1$  or  $\text{IT}(i) > NT$  for some  $i$  when  $NT \geq 2$ ,  
 or no value of  $\text{IT} = j$  for some  $j = 1, 2, \dots, NT$ , when  $NT \geq 2$ .

IFAIL = 4

On entry, the values of  $Y$  are constant.

IFAIL = 5

A computed standard error is zero due to rounding errors, or the eigenvalue computation failed to converge. Both are unlikely error exits.

IFAIL = 6

The treatments are totally confounded with blocks, so the treatment sum of squares and degrees of freedom are zero. The analysis of variance table is not computed, except for block and total sums of squares and degrees of freedom.

IFAIL = 7

The residual degrees of freedom or the residual sum of squares are zero, columns 3, 4 and 5 of the analysis of variance table will not be computed and the matrix of standard errors and covariances, C, will not be scaled by  $s$  or  $s^2$ .

IFAIL = 8

The design is disconnected; the standard errors may not be valid. The design may be nested.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL = -999

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

## 7 Accuracy

The algorithm used by G04BBF, described in Section 3, achieves greater accuracy than the traditional algorithms based on the subtraction of sums of squares.

## 8 Parallelism and Performance

G04BBF is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

G04BBF makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

## 9 Further Comments

To estimate missing values the Healy and Westmacott procedure or its derivatives may be used, see John and Quenouille (1977). This is an iterative procedure in which estimates of the missing values are adjusted by subtracting the corresponding values of the residuals. The new estimates are then used in the analysis of variance. This process is repeated until convergence. A suitable initial value may be the grand mean  $\hat{\mu}$ . When using this procedure IRDF should be set to the number of missing values plus one to obtain the correct degrees of freedom for the residual sum of squares.

For designs such as Graeco–Latin squares one or more of the blocking factors has to be removed in a preliminary analysis before the final analysis using calls to G04BBF or G04BCF. The residuals from the preliminary analysis are then input to G04BBF. In these cases IRDF should be set to the difference between N and the residual degrees of freedom from preliminary analysis. Care should be taken when using this approach as there is no check on the orthogonality of the two analyses.

For analysis of covariance the residuals are obtained from an analysis of variance of both the response variable and the covariates. The residuals from the response variable are then regressed on the residuals from the covariates using, say, G02CBF or G02DAF. The results from those routines can be used to test

for the significance of the covariates. To test the significance of the treatment effects after fitting the covariate, the residual sum of squares from the regression should be compared with the residual sum of squares obtained from the equivalent regression but using the residuals from fitting blocks only.

## 10 Example

The data, given by John and Quenouille (1977), are for a balanced incomplete block design with 10 blocks and 6 treatments and with 3 plots per block. The observations are the degree of pain experienced and the treatments are penicillin of different potency. The data is input and the analysis of variance table and treatment means are printed.

### 10.1 Program Text

```

Program g04bbfe

!      G04BBF Example Program Text

!      Mark 26 Release. NAG Copyright 2016.

!      .. Use Statements ..
Use nag_library, Only: g04bbf, nag_wp, x04caf
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Integer, Parameter          :: ldtabl = 4, nin = 5, nout = 6
!      .. Local Scalars ..
Real (Kind=nag_wp)         :: gmean, tol
Integer                    :: iblock, ifail, irdf, ldc, lit, n, nt
!      .. Local Arrays ..
Real (Kind=nag_wp), Allocatable :: bmean(:), c(:,,:), ef(:), r(:),      &
                                     tmean(:), wk(:), y(:)
Real (Kind=nag_wp)         :: tabl(ldtabl,5)
Integer, Allocatable       :: irep(:), it(:)
!      .. Intrinsic Procedures ..
Intrinsic                  :: abs
!      .. Executable Statements ..
Write (nout,*) 'G04BBF Example Program Results'
Write (nout,*)

!      Skip heading in data file
Read (nin,*)

!      Read in the problem size
Read (nin,*) n, nt, iblock

      ldc = nt
      If (nt>1) Then
         lit = n
      Else
         lit = 1
      End If
      Allocate (y(n),bmean(abs(iblock)),tmean(nt),irep(nt),c(ldc,nt),r(n),      &
               ef(nt),wk(3*nt),it(lit))

!      Read in the data and plot information
Read (nin,*) y(1:n)
If (nt>1) Then
   Read (nin,*) it(1:n)
End If

!      Use default tolerance
tol = 0.0E0_nag_wp

!      Use standard degrees of freedom
irdf = 0

!      Calculate the ANOVA table
ifail = 0

```

```

Call g04bbf(n,y,iblock,nt,it,gmean,bmean,tmean,tabl,ldtabl,c,ldc,irep,r, &
  ef,tol,irdf,wk,ifail)

!   Display results
Write (nout,*) ' ANOVA table'
Write (nout,*)
Write (nout,*) '   Source          df          SS          MS          F', &
  '   Prob'
Write (nout,*)
Write (nout,99998) ' Blocks          ', tabl(1,1:5)
Write (nout,99998) ' Treatments      ', tabl(2,1:5)
Write (nout,99998) ' Residual        ', tabl(3,1:3)
Write (nout,99998) ' Total           ', tabl(4,1:2)
Write (nout,*)
Write (nout,*) ' Efficiency Factors'
Write (nout,*)
Write (nout,99999) ef(1:nt)
Write (nout,*)
Write (nout,99997) ' Grand Mean', gmean
Write (nout,*)
Write (nout,*) ' Treatment Means'
Write (nout,*)
Write (nout,99999) tmean(1:nt)
Write (nout,*)
Flush (nout)
ifail = 0
Call x04caf('Lower','B',nt,nt,c,ldc,                                     &
  'Standard errors of differences between means',ifail)

99999 Format (8F10.2)
99998 Format (A,3X,F3.0,2X,3(F10.2,2X),F9.4)
99997 Format (A,F10.2)
End Program g04bbfe

```

## 10.2 Program Data

```

G04BBF Example Program Data
30 6 10           : N, NT, IBLOCK
1 5 4
5 10 6
2 9 3
4 8 6
2 4 7
6 7 5
5 7 2
7 2 4
8 4 2
10 8 7           : End of Y
1 2 3
1 2 4
1 3 5
1 4 6
1 5 6
2 3 6
2 4 5
2 5 6
3 4 5
3 4 6           : End of IT

```

## 10.3 Program Results

G04BBF Example Program Results

ANOVA table

Source	df	SS	MS	F	Prob
Blocks	9.	60.00	6.67	4.79	0.0039
Treatments	5.	101.78	20.36	14.62	0.0000
Residual	15.	20.89	1.39		



Total	29.	182.67				
Efficiency Factors						
0.00	0.80	0.80	0.80	0.80	0.80	0.80
Grand Mean	5.33					
Treatment Means						
2.50	7.25	8.08	5.92	2.92	5.33	
Standard errors of differences between means						
	1	2	3	4	5	6
1						
2	0.8344					
3	0.8344	0.8344				
4	0.8344	0.8344	0.8344			
5	0.8344	0.8344	0.8344	0.8344		
6	0.8344	0.8344	0.8344	0.8344	0.8344	

---